Final exam / 2008.12.12 / Calculus for Applications / MAT 3243.001

Note Title 12/12/2008

1. Sketch and label 5 level sets of $f(x, y) = x^2 - y^3$, including one at level 0.

```
> with(plots): with(linalg):
> x^2-y^3; contourplot(%,x=-3..3,y=-3..3,levels=[-2,-1,0,1,2],numpoints=4601);
```



2. In each case determine whether the limit exists, and if so, find the limit.

$$\text{(a)} \lim_{[x,y]\to 0} \frac{xy}{x^2+y^2} \qquad \text{(b)} \lim_{[x,y]\to 0} \frac{x^3-y^3}{x^2+xy+y^2}$$

a) If $x = 0$ we get $\frac{0}{y^2} \to 0$
if $x = y$ we get $\frac{x^2}{2x^2} \to \frac{1}{2}$
} not equal, so the limit does not exist

b) $$\frac{x^3-y^3}{x^2+xy+y^2} = \frac{(x-y)\cancel{(x^2+xy+y^2)}}{\cancel{x^2+xy+y^2}} = x - y \to \boxed{0}$$

3. If a trilobyte crawls south at 2 cm/s, it notices an increase in temperature at the rate of 1°/s. If it crawls west at 1 cm/s, the temperature increases by 3°/s. What is the rate of change of temperature if the cucaracha crawls southeast at 3 cm/s?

$$SE = \frac{1}{\sqrt{2}} \begin{bmatrix} 1 \\ -1 \end{bmatrix} \qquad \nabla T = \left[-3, -\frac{1}{2}\right]$$

$$\frac{dT}{dt} = \text{speed} \cdot D_{SE} T = \text{speed} \cdot \nabla T \cdot SE = 3 \left[-3, -\frac{1}{2}\right] \frac{1}{\sqrt{2}} \begin{bmatrix} 1 \\ -1 \end{bmatrix}$$

$$= \frac{3}{\sqrt{2}}\left[-3 + \frac{1}{2}\right] = -\frac{3 \cdot 5}{2\sqrt{2}} = \boxed{-\frac{15}{4}\sqrt{2} \text{ deg/s}}$$

4. Find the divergence and curl of $[x^2 y, \cos(xyz), y^2 z]$.

```
> F:=[x^2*y,cos(x*y*z),y^2*z];
```

$$F := [x^2 y, \cos(x y z), y^2 z]$$

```
> diverge(F,[x,y,z]); curl(F,[x,y,z]);
```

$$2 x y - \sin(x y z)\, x z + y^2$$

$$[2 y z + \sin(x y z)\, x y, 0, -\sin(x y z)\, y z - x^2]$$

5. Let $f = e^{x+y^2}$. Compute the Hessian matrix for f and find the quadratic Taylor approximation to f at the origin.

```
> f:=exp(x+y^2); subs({x=0,y=0},f): ff:=eval(%);
```

$$f := e^{(x+y^2)}$$

$$1$$

```
> jacobian([f],[x,y]); subs({x=0,y=0},%): df:=map(xx->eval(xx),%);
```

$$\begin{bmatrix} e^{(x+y^2)} & 2\,y\,e^{(x+y^2)} \end{bmatrix}$$

$$df := [1 \quad 0]$$

```
> hessian(f,[x,y]); subs({x=0,y=0},%): h:=map(xx->eval(xx),%);
```

$$\begin{bmatrix} e^{(x+y^2)} & 2\,y\,e^{(x+y^2)} \\ 2\,y\,e^{(x+y^2)} & 2\,e^{(x+y^2)} + 4\,y^2\,e^{(x+y^2)} \end{bmatrix}$$

$$h := \begin{bmatrix} 1 & 0 \\ 0 & 2 \end{bmatrix}$$

```
> v:=[x,y];
```

$$v := [x, y]$$

```
> ff+df&*v+(1/2)*transpose(v)&*h&*v: evalm(%);
```

$$\left[x + 1 + \frac{1}{2}x^2 + y^2\right]$$

Check: $e^t = 1 + t + \frac{1}{2}t^2 + \dots$

$e^{x+y^2} = 1 + x + y^2 + \frac{1}{2}(x+y^2)^2 + \dots = 1 + x + y^2 + \frac{1}{2}x^2 + \dots$

6. A solid is bounded by the coordinate planes and the plane $x + 2y + 7z = 14$. Set up, but do not evaluate the iterated integral for the volume with the order of integration y, x, z.

```
> p:=x+2*y+7*z=14;
```

$$p := x + 2\,y + 7\,z = 14$$

```
> implicitplot3d(p,x=0..14,y=0..7,z=0..2,axes=normal,style=patchnogrid,scaling=constrained);
```



$$\int_0^2 \left[\int_0^{14-7z} \left[\int_0^{7-\frac{1}{2}x-\frac{7}{2}z} dy\right] dx\right] dz$$

7. Integrate $\omega = x\,dx + y\,dy$ along the straight line segment from $[-1,-1]$ to $[1,1]$. Had we chosen a different path from $[-1,-1]$ to $[1,1]$, would the integral remain the same? Explain.

```
> [-1,-1]*(1-t)+[1,1]*t; X:=evalm(%);
```

$$[-1, -1](1-t) + [1, 1]\,t$$

$$X := [-1 + 2\,t, -1 + 2\,t]$$

```
> dX:=map(xx->diff(xx,t),X);
```

$$dX := [2, 2]$$

```
> omega:=X[1]*dX[1]+X[2]*dX[2];
```

$$\omega := -4 + 8\,t$$

```
> int(omega,t); int(omega,t=0..1);
```

$$-4\,t + 4\,t^2$$

$$0$$

$d\omega = dx\,dx + dy\,dy = 0$, so the integral is path independent.

8. Find first a parametric formula and then an equation for the plane in $\mathbf{R}^3$ tangent to the surface $[s+t, st, \sin(st)]$ at $[1,0,0]$.

```
> X:=[s+t,s*t,sin(s*t)];
```

$$X := [s+t,\, s\,t,\, \sin(s\,t)]$$

```
> p:=[1,0]; subs({s=p[1],t=p[2]},X): XX:=eval(%);
```

$$p := [1, 0]$$

$$XX := [1, 0, 0]$$

```
> jacobian(X,[s,t]); subs({s=p[1],t=p[2]},%): dX:=map(xx->eval(xx),%);
```

$$\begin{bmatrix} 1 & 1 \\ t & s \\ \cos(s\,t)\,t & \cos(s\,t)\,s \end{bmatrix}$$

$$dX := \begin{bmatrix} 1 & 1 \\ 0 & 1 \\ 0 & 1 \end{bmatrix}$$

```
> v:=[s-p[1],t-p[2]];
```

$$v := [s-1, t]$$

```
> XX+dX&*v: param:=evalm(%);
```

$$param := [s+t, t, t]$$

```
> p1:=plot3d(X,s=0..2,t=-1..1,style=patchnogrid):
> p2:=plot3d(param,s=0..2,t=-1..1,style=patchnogrid,color=red):
> display({p1,p2},axes=frame);
```




```
> N:=crossprod(col(dX,1),col(dX,2));
```

$$N := [0, -1, 1]$$

```
> plane:=dotprod([x,y,z]-XX,N)=0;
```

$$plane := -y+z=0$$

```
> p3:=implicitplot3d(plane,x=0..2,y=-1..1,z=-1..1,style=patchnogrid,color=blue):
> display({p1,p3},axes=frame);
```



9. Parametrize the paraboloid $z = 2 - x^2 - y^2, z \geq 1$ oriented with the upward normal. Compute the flux of $\mathbf{F} = [x, y, -2z]$ through this surface. Would the flux of $\mathbf{F}$ through the unit disc in the $z = 1$ plane differ? Explain.

```
> [x,y,1]; subs({x=r*cos(theta),y=r*sin(theta)},%): X:=simplify(%);
```

$$[x, y, 1]$$

$$X := [r\cos(\theta), r\sin(\theta), 1]$$

```
> [x,y,2-x^2-y^2]; subs({x=r*cos(theta),y=r*sin(theta)},%): X:=simplify(%);
```

$$[x, y, 2 - x^2 - y^2]$$

$$X := [r\cos(\theta), r\sin(\theta), 2 - r^2]$$

```
> plot3d(X,r=0..1,theta=-Pi..Pi,axes=frame);
```




```
> dX:=jacobian(X,[r,theta]);
```

$$dX := \begin{bmatrix} \cos(\theta) & -r\sin(\theta) \\ \sin(\theta) & r\cos(\theta) \\ -2r & 0 \end{bmatrix}$$

```
> crossprod(col(dX,1),col(dX,2)); N:=simplify(%);
```

$$[2r^2\cos(\theta), 2r^2\sin(\theta), \cos(\theta)^2 r + \sin(\theta)^2 r]$$

$$N := [2r^2\cos(\theta), 2r^2\sin(\theta), r]$$

```
> [x,y,-2*z]; subs({x=X[1],y=X[2],z=X[3]},%): F:=simplify(%);
```

$$[x, y, -2z]$$

$$F := [r\cos(\theta), r\sin(\theta), -4 + 2r^2]$$

```
> sum(F[i]*N[i],i=1..3); simplify(%); int(%,r=0..1)*2*Pi;
```

$$2r^3\cos(\theta)^2 + 2r^3\sin(\theta)^2 + (-4 + 2r^2)r$$

$$4r(r^2 - 1)$$

$$-2\pi$$

Since $\operatorname{div} F = 0$, flux through another surface with the same boundary will be the same.